Appendix A2. Picard Iteration

The following first order initial value problem

$$y'(x) = f(x, y(x)), \quad y(0) = y0$$

can be reformulated as the an integral equation

$$y(x) = y0 + \int_{x0}^{x} f(t, y(t))\, dt\,.$$

See Simmons/Krantz, Chapter 3, Section 2.2.

Assume that $f(x, y)$ and its y partial are continuous on a rectangle containing the point $(x0, y0)$. Then starting with an initial "guess" for a solution

$$y = y_0(x)$$

the recursive relation

$$y_{j+1}(t) = y0 + \int_{x0}^{x} f\left(t, y_j(t)\right) dt$$

generates a sequence of functions converging to the solution. These functions are called the *Picard iterates* generated by $y_0(x)$

Maple can generate some of these iterates, use a **for..do** loop.

The following example is Example 3.16 in Simmons/Krantz.

Example Obtain the first three Picard iterates for the following IVP.

$$y' = 2\,y\ , \qquad y(0) = 1$$

Sketch their graphs and the graph of the exact solution.

Note that the first approximation should be $y_0(x) = 1$. We begin with the definition of the function f.

```
> f := (x,y) -> 2*y;
```

$$f := (x, y) \rightarrow 2\,y \tag{1}$$

Use it to define the differential equation.

```
> DE := diff(y(x),x) = f(x,y(x));
```

$$DE := \frac{d}{dx}\, y(x) = 2\,y(x) \tag{2}$$

Compute the first three iterates. We use Y_j to denote the jth approximation function.

```
> Y[0] := x -> 1:
  for n from 1 to 3
    do
      Y[0](x) + int(f(t,Y[n-1](t)),t=0..x):
      Y[n] := unapply(%,x):
    end do:
```

```
unassign('n');
```

The formulas for the initial guess and the three iterates are displayed below.

```
> for n from 0 to 3
    do
      y[n](x) = Y[n](x)
    end do;
  unassign('n');
```

$$y_0(x) = 1$$

$$y_1(x) = 1 + 2x$$

$$y_2(x) = 1 + 2x + 2x^2$$

$$y_3(x) = 1 + 2x + 2x^2 + \frac{4}{3}x^3 \tag{3}$$

The actual solution formula is obtained next.

```
> soln := dsolve( {DE, y(0)=1} );
```

$$soln := y(x) = e^{2x} \tag{4}$$

The plot of the solution, red and thick, and the three iterates, blue, green, black:

```
> plots[setoptions]( tickmarks = [5,5] );
> plot( [rhs(soln),Y[n](x)$n=1..3], x=0..2, y=0..5,
         color=[red,blue,green,black], thickness=[2,1$3]);
```



The next output reveals that the Picard iterates are the partial sums to the Taylor series for the solution.

```
> dsolve( {DE, y(0)=1}, y(x), type=series);
```

$$y(x) = \left(1 + 2x + 2x^2 + \frac{4}{3}x^3 + \frac{2}{3}x^4 + \frac{4}{15}x^5 + O\left(x^6\right)\right) \tag{5}$$